
ExCmd is an Elixir library for running and communicating with external programs using a back-pressure mechanism. It provides a robust alternative to Elixir's built-in Port with improved memory management through demand-driven I/O.
The Port I/O Problem
When using Elixir's built-in Port, running external programs that generate large amounts of output (like streaming video using ffmpeg) can quickly lead to memory issues. This happens because Port I/O is not demand-driven - it consumes output from stdout as soon as it's available and sends it to the process mailbox. Since BEAM process mailboxes are unbounded, the output accumulates there waiting to be received.
Memory Usage Comparison
Let's look at how ExCmd handles memory compared to Port when processing large streams:
Using Port (memory grows unbounded):
Port.open({:spawn_executable, "/bin/cat"}, [{:args, ["/dev/random"]}, {:line, 10}, :binary, :use_stdio])
[image: Port memory consumption]
Using ExCmd (memory remains stable):
ExCmd.stream!(~w(cat /dev/random))
|> Enum.each(fn data ->
  IO.puts(IO.iodata_length(data))
end)
[image: ExCmd memory consumption]
ExCmd solves this by implementing:
	Demand-driven I/O with proper back-pressure
	Efficient use of OS-backed stdio buffers
	Stream-based API that integrates with Elixir's ecosystem

Key Features
	Back-pressure Support: Controls data flow between your application and external programs
	Stream Abstraction: Seamless integration with Elixir's Stream API
	Memory Efficient: Demand-driven I/O prevents memory issues with large data transfers
	Cross-platform: Pre-built binaries for MacOS, Windows, and Linux
	Process Management: Proper program termination with no zombie processes
	Selective I/O Control: Ability to close stdin while keeping stdout open
	No Dependencies: No separate middleware or shim installation required

Installation
Add ex_cmd to your list of dependencies in mix.exs:
def deps do
  [
    {:ex_cmd, "~> x.x.x"}
  ]
end
Quick Start Examples
Basic Command Execution
# Simple command execution
ExCmd.stream!(~w(echo Hello))
|> Enum.into("")
# => "Hello\n"

# Get your IP address
ExCmd.stream!(~w(curl ifconfig.co))
|> Enum.into("")
Working with Input Streams
# String input
ExCmd.stream!(~w(cat), input: "Hello World")
|> Enum.into("")
# => "Hello World"

# List of strings
ExCmd.stream!(~w(cat), input: ["Hello", " ", "World"])
|> Enum.into("")
# => "Hello World"

# Binary data
ExCmd.stream!(~w(base64), input: <<1, 2, 3, 4, 5>>)
|> Enum.into("")
# => "AQIDBAU=\n"

# IOData
ExCmd.stream!(~w(base64), input: [<<1, 2>>, [3], [<<4, 5>>]])
|> Enum.into("")
# => "AQIDBAU=\n"
Media Processing Examples
# Extract audio from video with controlled memory usage
ExCmd.stream!(~w(ffmpeg -i pipe:0 -f mp3 pipe:1),
  input: File.stream!("music_video.mkv", [], 65536))
|> Stream.into(File.stream!("music.mp3"))
|> Stream.run()

# Process video streams efficiently
ExCmd.stream!(~w(ffmpeg -i pipe:0 -c:v libx264 -f mp4 pipe:1),
  input: File.stream!("input.mp4", [], 65536),
  max_chunk_size: 65536)
|> Stream.into(File.stream!("output.mp4"))
|> Stream.run()
Error Handling
# stream!/2 raises on non-zero exit status
ExCmd.stream!(["sh", "-c", "exit 10"])
|> Enum.to_list()
# => ** (ExCmd.Stream.AbnormalExit) program exited with exit status: 10

# stream/2 returns exit status as last element
ExCmd.stream(["sh", "-c", "echo 'foo' && exit 10"])
|> Enum.to_list()
# => ["foo\n", {:exit, {:status, 10}}]
Advanced Features
# Redirect stderr to stdout
ExCmd.stream!(["sh", "-c", "echo foo; echo bar >&2"],
  stderr: :redirect_to_stdout)
|> Enum.into("")
# => "foo\nbar\n"
Alternatives
	For NIF-based solutions without middleware overhead, consider Exile
	For simple command execution without streaming, Elixir's built-in Port might be sufficient

Documentation
Detailed documentation is available at HexDocs.
License
See LICENSE file for details.


  

    LICENSE


MIT License

Copyright (c) 2020 Akash Hiremath

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.



  

    
ExCmd 
    



      
ExCmd is an alternative for beam ports
with back-pressure and non-blocking IO.
Quick Start
Run a command and read from stdout
iex> ExCmd.stream!(~w(echo Hello))
...> |> Enum.into("") # collect as string
"Hello\n"
String input
iex> ExCmd.stream!(~w(cat), input: "Hello World")
iex> |> Enum.into("")
"Hello World"
List of strings
iex> ExCmd.stream!(~w(cat), input: ["Hello", " ", "World"])
iex> |> Enum.into("")
"Hello World"
Binary data
iex> ExCmd.stream!(~w(base64), input: <<1, 2, 3, 4, 5>>)
iex> |> Enum.into("")
"AQIDBAU=\n"
IOData
iex> ExCmd.stream!(~w(base64), input: [<<1, 2>>, [3], [<<4, 5>>]])
iex> |> Enum.into("")
"AQIDBAU=\n"
Run a command with list of strings as input
iex> ExCmd.stream!(~w(cat), input: ["Hello", " ", "World"])
...> |> Enum.into("") # collect as string
"Hello World"
Run a command with input as Stream
iex> input_stream = Stream.map(1..10, fn num -> "#{num} " end)
iex> ExCmd.stream!(~w(cat), input: input_stream)
...> |> Enum.into("")
"1 2 3 4 5 6 7 8 9 10 "
Run a command with input as infinite stream
# create infinite stream
iex> input_stream = Stream.repeatedly(fn -> "A" end)
iex> binary =
...>   ExCmd.stream!(~w(cat), input: input_stream, ignore_epipe: true) # we need to ignore epipe since we are terminating the program before the input completes
...>   |> Stream.take(2) # we must limit since the input stream is infinite
...>   |> Enum.into("")
iex> is_binary(binary)
true
iex> "AA" <> _ = binary # we might get more than 2 "A" due to buffering
Run a command with input Collectable
# ExCmd calls the callback with a sink where the process can push the data
iex> ExCmd.stream!(~w(cat), input: fn sink ->
...>   Stream.map(1..10, fn num -> "#{num} " end)
...>   |> Stream.into(sink) # push to the external process
...>   |> Stream.run()
...> end)
...> |> Stream.take(100) # we must limit since the input stream is infinite
...> |> Enum.into("")
"1 2 3 4 5 6 7 8 9 10 "
When the command wait for the input stream to close
# base64 command wait for the input to close and writes data to stdout at once
iex> ExCmd.stream!(~w(base64), input: ["abcdef"])
...> |> Enum.into("")
"YWJjZGVm\n"
stream!/2 raises non-zero exit as error
iex> ExCmd.stream!(["sh", "-c", "echo 'foo' && exit 10"])
...> |> Enum.to_list()
** (ExCmd.Stream.AbnormalExit) program exited with exit status: 10
stream/2 variant returns exit status as last element
iex> ExCmd.stream(["sh", "-c", "echo 'foo' && exit 10"])
...> |> Enum.to_list()
[
  "foo\n",
  {:exit, {:status, 10}} # returns exit status of the program as last element
]
You can fetch exit_status from the error for stream!/2
iex> try do
...>   ExCmd.stream!(["sh", "-c", "exit 10"])
...>   |> Enum.to_list()
...> rescue
...>   e in ExCmd.Stream.AbnormalExit ->
...>     e.exit_status
...> end
10
With max_chunk_size set
iex> data =
...>   ExCmd.stream!(~w(cat /dev/urandom), max_chunk_size: 100, ignore_epipe: true)
...>   |> Stream.take(5)
...>   |> Enum.into("")
iex> byte_size(data)
500
When input and output run at different rate
iex> input_stream = Stream.map(1..1000, fn num -> "X #{num} X\n" end)
iex> ExCmd.stream!(~w(grep 250), input: input_stream)
...> |> Enum.into("")
"X 250 X\n"
With stderr set to :disable (default)
iex> ExCmd.stream!(["sh", "-c", "echo foo; echo bar >&2"])
...> |> Enum.to_list()
["foo\n"]
With stderr set to :redirect_to_stdout
iex> ExCmd.stream!(["sh", "-c", "echo foo; echo bar >&2"], stderr: :redirect_to_stdout)
...> |> Enum.into("")
"foo\nbar\n"
With stderr set to :console
ExCmd.stream!(["sh", "-c", "echo foo; echo bar >&2"], stderr: :console)
|> Enum.to_list()
# STDERR output will be written to the console
For more details about stream API, see ExCmd.stream!/2 and ExCmd.stream/2.
For more details about inner working, please check ExCmd.Process
documentation.

          @spec stream([String.t(), ...],
  input: Enum.t() | collectable_func() | String.t() | binary(),
  exit_timeout: timeout(),
  cd: String.t(),
  env: [{String.t(), String.t()}],
  stderr: :console | :redirect_to_stdout | :disable,
  ignore_epipe: boolean(),
  max_chunk_size: pos_integer()
) :: ExCmd.Stream.t()


      


Same as ExCmd.stream!/2 but the program exit status is passed as last
element of the stream.
The last element will be of the form {:exit, term()}. term will be a
two-element tuple with :status and a positive integer in case of normal exit
(e.g. {:status, 0} or {:status, 2}), and :epipe in case of epipe error.
See ExCmd.stream!/2 documentation for details about the options and
examples.

          @spec stream!([String.t(), ...],
  input: Enum.t() | collectable_func() | String.t() | binary(),
  exit_timeout: timeout(),
  cd: String.t(),
  env: [{String.t(), String.t()}],
  stderr: :console | :redirect_to_stdout | :disable,
  ignore_epipe: boolean(),
  max_chunk_size: pos_integer()
) :: ExCmd.Stream.t()


      


Runs the command with arguments and return an the stdout as lazily
Enumerable stream, similar to Stream.
First parameter must be a list containing command with arguments.
Example: ["cat", "file.txt"].
Options
	input - Input can be either an Enumerable or a function which accepts Collectable.
	String or Binary:
# List
ExCmd.stream!(~w(cat), input: "Hello World") |> Enum.into("")
# Stream
ExCmd.stream!(~w(cat), input: <<1, 2, 3, 4, 5>>) |> Enum.into("")

	Enumerable:
# List
ExCmd.stream!(~w(base64), input: ["hello", "world"]) |> Enum.to_list()
# Stream
ExCmd.stream!(~w(cat), input: File.stream!("log.txt", [], 65_531)) |> Enum.to_list()

	Collectable:
If the input in a function with arity 1, ExCmd will call that function
with a Collectable as the argument. The function must push input to this
collectable. Return value of the function is ignored.
ExCmd.stream!(~w(cat), input: fn sink -> Enum.into(1..100, sink, &to_string/1) end)
|> Enum.to_list()
By defaults no input is sent to the command.



	exit_timeout - Duration to wait for external program to exit after completion
(when stream ends). Defaults to :infinity

	max_chunk_size - Maximum size of iodata chunk emitted by the stream.
Chunk size can be less than the max_chunk_size depending on the amount of
data available to be read. Defaults to 65_531. Value must <= 65_531 and > 0.

	stderr  -  different ways to handle stderr stream.
	:console  -  stderr output is redirected to console (Default)
	:redirect_to_stdout  -  stderr output is redirected to stdout
	:disable  -  stderr output is redirected /dev/null suppressing all output

 See :stderr for more details and issues associated with them.

	ignore_epipe - When set to true, reader can exit early without raising error.
Typically writer gets EPIPE error on write when program terminate prematurely.
With ignore_epipe set to true this error will be ignored. This can be used to
match UNIX shell default behaviour. EPIPE is the error raised when the reader finishes
the reading and close output pipe before command completes. Defaults to false.


Remaining options are passed to ExCmd.Process.start_link/2
If program exits with non-zero exit status or :epipe then ExCmd.Stream.AbnormalExit
error will be raised with exit_status field set.
Examples
ExCmd.stream!(~w(ffmpeg -i pipe:0 -f mp3 pipe:1), input: File.stream!("music_video.mkv", [], 65_535))
|> Stream.into(File.stream!("music.mp3"))
|> Stream.run()
Stream with stderr redirected to stdout
ExCmd.stream!(["sh", "-c", "echo foo; echo bar >&2"], stderr: :redirect_to_stdout)
|> Stream.map(&IO.write/1)
|> Stream.run()

GenServer which wraps spawned external command.
Use ExCmd.stream!/1 over using this. Use this only if you are
familiar with life-cycle and need more control of the IO streams
and OS process.
Comparison with Port
	it is demand driven. User explicitly has to read the command
output, and the progress of the external command is controlled
using OS pipes. ExCmd never load more output than we can consume,
so we should never experience memory issues

	it can close stdin while consuming output

	tries to handle zombie process by attempting to cleanup
external process. Note that there is no middleware involved
with ex_cmd so it is still possible to endup with zombie process.

	selectively consume stdout and stderr


Internally ExCmd uses non-blocking asynchronous system calls
to interact with the external process. It does not use port's
message based communication, instead uses raw stdio and NIF.
Uses asynchronous system calls for IO. Most of the system
calls are non-blocking, so it should not block the beam
schedulers. Make use of dirty-schedulers for IO
Introduction
ExCmd.Process is a process based wrapper around the external
process. It is similar to port as an entity but the interface is
different. All communication with the external process must happen
via ExCmd.Process interface.
ExCmd process life-cycle tied to external process and owners. All
system resources such are open file-descriptors, external process
are cleaned up when the ExCmd.Process dies.
Owner
Each ExCmd.Process has an owner. And it will be the process which
created it (via ExCmd.Process.start_link/2). Process owner can not
be changed.
Owner process will be linked to the ExCmd.Process. So when the
ex_cmd process is dies abnormally the owner will be killed too or
visa-versa. Owner process should avoid trapping the exit signal, if
you want avoid the caller getting killed, create a separate process
as owner to run the command and monitor that process.
Only owner can get the exit status of the command, using
ExCmd.Process.await_exit/2. All ex_cmd processes MUST be
awaited.  Exit status or reason is ALWAYS sent to the owner. It
is similar to Task. If the
owner exit without await_exit, the ex_cmd process will be killed,
but if the owner continue without await_exit then the ex_cmd
process will linger around till the process exit.
iex> alias ExCmd.Process
iex> {:ok, p} = Process.start_link(~w(echo hello))
iex> Process.read(p, 100)
{:ok, "hello\n"}
iex> Process.read(p, 100) # read till we get :eof
:eof
iex> Process.await_exit(p)
{:ok, 0}
Pipe & Pipe Owner
Standard IO pipes/channels/streams of the external process such as
STDIN, STDOUT, STDERR are called as Pipes. User can either write or
read data from pipes.
Each pipe has an owner process and only that process can write or
read from the ex_cmd process. By default the process who created the
ex_cmd process is the owner of all the pipes. Pipe owner can be
changed using ExCmd.Process.change_pipe_owner/3.
Pipe owner is monitored and the pipes are closed automatically when
the pipe owner exit. Pipe Owner can close the pipe early using
ExCmd.Process.close_stdin/1 etc.
ExCmd.Process.await_exit/2 closes all of the caller owned pipes by
default.
iex> {:ok, p} = Process.start_link(~w(cat))
iex> writer = Task.async(fn ->
...>   :ok = Process.change_pipe_owner(p, :stdin, self())
...>   Process.write(p, "Hello World")
...> end)
iex> Task.await(writer)
:ok
iex> Process.read(p, 100)
{:ok, "Hello World"}
iex> Process.await_exit(p)
{:ok, 0}
Pipe Operations
Only Pipe owner can read or write date to the owned pipe.
All Pipe operations (read/write) blocks the caller as a mechanism
to put back-pressure, and this also makes the API simpler.
This is same as how command-line programs works on the shell,
along with pipes in-between, Example: cat larg-file | grep "foo".
Internally ExCmd uses asynchronous IO APIs to avoid blocking VM
(by default NIF calls blocks the VM scheduler),
so you can open several pipes and do concurrent IO operations without
blocking VM.
stderr
by default is :stderr is connected to console, data written to
stderr will appear on the console.
You can change the behavior by setting :stderr:
	:console  -  stderr output is redirected to console
	:redirect_to_stdout  -  stderr output is redirected to stdout
	:disable  -  stderr output is redirected /dev/null suppressing all output. See below for more details. (Default)

Using redirect_to_stdout
stderr data will be redirected to stdout. When you read stdout
you will see both stdout & stderr combined and you won't be
able differentiate stdout and stderr separately.
This is similar to :stderr_to_stdout option present in
Ports.
Unexpected Behaviors
On many systems, stdout and stderr are separated. And between
the source program to ExCmd, via the kernel, there are several places
that may buffer data, even temporarily, before ExCmd is ready
to read them. There is no enforced ordering of the readiness of
these independent buffers for ExCmd to make use of.
This can result in unexpected behavior, including:
	mangled data, for example, UTF-8 characters may be incomplete
until an additional buffered segment is released on the same
source
	raw data, where binary data sent on one source, is incompatible
with data sent on the other source.
	interleaved data, where what appears to be synchronous, is not

In short, the two streams might be combined at arbitrary byte position
leading to above mentioned issue.
Most well-behaved command-line programs are unlikely to exhibit
this, but you need to be aware of the risk.
A good example of this unexpected behavior is streaming JSON from
an external tool to ExCmd, where normal JSON output is expected on
stdout, and errors or warnings via stderr. In the case of an
unexpected error, the stdout stream could be incomplete, or the
stderr message might arrive before the closing data on the stdout
stream.
Process Termination
When owner does (normally or abnormally) the ExCmd process always
terminated irrespective of pipe status or process status. External
process get a chance to terminate gracefully, if that fail it will
be killed.
If owner calls await_exit then the owner owned pipes are closed
and we wait for external process to terminate, if the process
already terminated then call returns immediately with exit
status. Else command will be attempted to stop gracefully following
the exit sequence based on the timeout value (5s by default).
If owner calls await_exit with timeout as :infinity then
ExCmd does not attempt to forcefully stop the external command and
wait for command to exit on itself. The await_exit call can be blocked
indefinitely waiting for external process to terminate.
If external process exit on its own, exit status is collected and
ExCmd process will wait for owner to close pipes. Most commands exit
with pipes are closed, so just ensuring to close pipes when works is
done should be enough.
Example of process getting terminated by SIGTERM signal
# sleep command does not watch for stdin or stdout, so closing the
# pipe does not terminate the sleep command.
iex> {:ok, p} = Process.start_link(~w(sleep 100000000)) # sleep indefinitely
iex> Process.await_exit(p, 2000) # ensure `await_exit` finish within `2000ms`. By default it waits for 5s
{:error, :killed} # command exit due to SIGTERM
Examples
Run a command without any input or output
iex> {:ok, p} = Process.start_link(["sh", "-c", "exit 2"])
iex> Process.await_exit(p)
{:ok, 2}
Single process reading and writing to the command
# bc is a calculator, which reads from stdin and writes output to stdout
iex> {:ok, p} = Process.start_link(~w(cat))
iex> Process.write(p, "hello\n") # there must be new-line to indicate the end of the input line
:ok
iex> Process.read(p)
{:ok, "hello\n"}
iex> Process.write(p, "world\n")
:ok
iex> Process.read(p)
{:ok, "world\n"}
# We must close stdin to signal the command that we are done.
# since `await_exit` implicitly closes the pipes, in this case we don't have to
iex> Process.await_exit(p)
{:ok, 0}
Running a command which flush the output on stdin close. This is not
supported by Erlang/Elixir ports.
# `base64` command reads all input and writes encoded output when stdin is closed.
iex> {:ok, p} = Process.start_link(~w(base64))
iex> Process.write(p, "abcdef")
:ok
iex> Process.close_stdin(p) # we can selectively close stdin and read all output
:ok
iex> Process.read(p)
{:ok, "YWJjZGVm\n"}
iex> Process.read(p) # typically it is better to read till we receive :eof when we are not sure how big the output data size is
:eof
iex> Process.await_exit(p)
{:ok, 0}
Read and write to pipes in separate processes
iex> {:ok, p} = Process.start_link(~w(cat))
iex> writer = Task.async(fn ->
...>   :ok = Process.change_pipe_owner(p, :stdin, self())
...>   Process.write(p, "Hello World")
...>   # no need to close the pipe explicitly here. Pipe will be closed automatically when process exit
...> end)
iex> reader = Task.async(fn ->
...>   :ok = Process.change_pipe_owner(p, :stdout, self())
...>   Process.read(p)
...> end)
iex> :timer.sleep(500) # wait for the reader and writer to change pipe owner, otherwise `await_exit` will close the pipes before we change pipe owner
iex> Process.await_exit(p, :infinity) # let the reader and writer take indefinite time to finish
{:ok, 0}
iex> Task.await(writer)
:ok
iex> Task.await(reader)
{:ok, "Hello World"}

        await_exit(process, timeout \\ 5000)

      


        Wait for the program to terminate and get exit status.

        Closes external program's standard error pipe (stderr)



    


    
      
        close_stdin(process)

      


        Closes external program's standard input pipe (stdin).

        Closes external program's standard output pipe (stdout)

        Returns bytes from executed command's stdout with maximum size max_size.

          @type exit_status() :: non_neg_integer()

          @type pipe_name() :: :stdin | :stdout | :stderr

          @type recv_commands() :: :output | :output_eof | :send_input | :exit_status

          @type t() :: %ExCmd.Process{
  exit_ref: reference(),
  monitor_ref: reference(),
  owner: pid(),
  pid: pid() | nil
}

          @spec await_exit(t(), timeout :: timeout()) ::
  {:ok, exit_status()} | {:error, :killed} | {:error, term()}


      


Wait for the program to terminate and get exit status.
ONLY the Process owner can call this function. And all ExCmd
process MUST be awaited (Similar to Task).
ExCmd first politely asks the program to terminate by closing the
pipes owned by the process owner (by default process owner is the
pipes owner). Most programs terminates when standard pipes are
closed.
If you have changed the pipe owner to other process, you have to
close pipe yourself or wait for the program to exit.
If the program fails to terminate within the timeout (default 5s)
then the program will be killed using the exit sequence by sending
SIGTERM, SIGKILL signals in sequence.
When timeout is set to :infinity await_exit wait for the
programs to terminate indefinitely.
For more details check module documentation.

          @spec change_pipe_owner(t(), pipe_name(), pid()) :: :ok | {:error, any()}


      


Changes the Pipe owner of the pipe to specified pid.
Note that currently any process can change the pipe owner.
For more details about Pipe Owner, please check module docs.

Returns a specification to start this module under a supervisor.
See Supervisor.

          @spec close_stderr(t()) :: :ok | {:error, any()}


      


Closes external program's standard error pipe (stderr)
Only owner of the pipe can close the pipe. This call will return
immediately.

  



  
    
      
    
    
      close_stdin(process)



        
          
        

    

  


  

      

          @spec close_stdin(t()) ::
  :ok | {:error, :pipe_closed_or_invalid_caller} | {:error, any()}


      


Closes external program's standard input pipe (stdin).
Only owner of the pipe can close the pipe. This call will return
immediately.

  



  
    
      
    
    
      close_stdout(process)



        
          
        

    

  


  

      

          @spec close_stdout(t()) :: :ok | {:error, any()}


      


Closes external program's standard output pipe (stdout)
Only owner of the pipe can close the pipe. This call will return
immediately.

  



  
    
      
    
    
      os_pid(process)



        
          
        

    

  


  

      

          @spec os_pid(t()) :: pos_integer()


      


Returns OS pid of the command
This is meant only for debugging. Avoid interacting with the
external process directly

  



    

  
    
      
    
    
      read(process, max_size \\ 65531)



        
          
        

    

  


  

      

          @spec read(t(), pos_integer()) :: {:ok, iodata()} | :eof | {:error, any()}


      


Returns bytes from executed command's stdout with maximum size max_size.
Blocks if no data present in stdout pipe yet. And returns as soon as
data of any size is available.
Note that max_size is the maximum size of the returned data. But
the returned data can be less than that depending on how the program
flush the data etc.

  



    

  
    
      
    
    
      start_link(list, opts \\ [])



        
          
        

    

  


  

      

          @spec start_link([String.t(), ...],
  cd: String.t(),
  env: [{String.t(), String.t()}],
  stderr: :console | :redirect_to_stdout | :disable
) :: {:ok, t()} | {:error, any()}


      


Starts ExCmd.Process server.
Starts external program using cmd_with_args with options opts
cmd_with_args must be a list containing command with arguments.
example: ["cat", "file.txt"].
Options
	cd   -  the directory to run the command in

	env  -  a list of tuples containing environment key-value.
These can be accessed in the external program

	stderr  -  different ways to handle stderr stream.
	:console  -  stderr output is redirected to console
	:redirect_to_stdout  -  stderr output is redirected to stdout
	:disable  -  stderr output is redirected /dev/null suppressing all output (Default)

See :stderr for more details and issues associated with them


Caller of the process will be the owner owner of the ExCmd Process.
And default owner of all opened pipes.
Please check module documentation for more details

  



  
    
      
    
    
      write(process, iodata)



        
          
        

    

  


  

      

          @spec write(t(), binary()) :: :ok | {:error, any()}


      


Writes iodata data to external program's standard input pipe.
This call blocks when the pipe is full. Returns :ok when
the complete data is written.

  


        

      


  

    
ExCmd.Stream 
    



      
Defines a ExCmd.Stream struct returned by ExCmd.stream!/2.

      


      
        Summary


  
    Types
  


    
      
        t()

      


        Struct members are private, do not depend on them



    





      


      
        Types


        


  
    
      
    
    
      t()



        
          
        

    

  


  

      

          @type t() :: %ExCmd.Stream{
  cmd_with_args: [String.t()],
  process_opts: keyword(),
  stream_opts: map()
}


      


Struct members are private, do not depend on them

  


        

      


  

    
ExCmd.Process.Error exception
    






  

    
ExCmd.Stream.AbnormalExit exception

